Using Mathematica

This is a tutorial and exercise to familiarize you with Mathematica 3.0. The current version is 6.0. If you use a newer version there may be some differences. I would like to make it clear that although *Mathematica* is a powerful tool that will simplify your life, you should work hard to strengthen and maintain your mathematical and analytical skills. Otherwise, you may enter something incorrectly and not even realize that *Mathematica* has given you the wrong answer.

■ General

To write equations, it is easiest to use the Mathematica palettes.Eventually, you can learn and implement keyboard commands instead of using the mouse.The " Basic Input " palette is probably the most useful.If it is not showing, you can make it appear by choosing FILE/PALETTES/BasicInput.You should notice that there are many other palettes that can be used.You should examine them so that you will know what is available.

■ Basic Palette and Commands

For fractions, you can click the fraction button on the " Basic Input " palette, which will give you $\frac{\square}{\square}$. You can fill in the numerator and denominator by clicking or tabbing and entering the values. The keyboard shortcut is the "Control" key ([CTRL]) and the "/" key at the same time (I will denote this type of double key stroke like this [CTRL]-[/]).

Special characters can be entered using the "Basic Input" palette or other Typesetting and Character palettes in the File/Palettes menu. For example, you can click on the π symbol on the "Basic Input" palette. The keyboard shortcut is to hit the "esc" key, type "pi", and hit the "esc" key again. For many characters, this escape, spell, escape technique will work. Here is a short list of examples. The quotations show you strings to type, but the quotations, themselves, are not typed.

π	[ESC], "pi", [ESC]
ρ	[ESC], "rho", [ESC]
δ	[ESC], "delta", [ESC]
Δ	[ESC], "Delta", [ESC]
$\sqrt{}$	[ESC], "sqrt", [ESC]
10^{23}	"10", [CTRL]-[6], "23"
f_i	"f", [CTRL]-[-], "i"
$\sum$	[ESC], "sum", [ESC]
Σ	[ESC], "Sigma", [ESC]
$\int$	[ESC], "int", [ESC]

■ Numerical Calculations

To perform a calculation, use the ENTER key on the number pad or type [SHIFT]-[ENTER] (shift and enter at the same time). Just typing [ENTER] will carriage return for a new line of input without evaluating what you have entered.

You can do arithmetic such as

$$\frac{5 + 8}{\pi}$$

$$\frac{13}{\pi}$$

You can have the output be decimal form with as many decimals as you like by using the N[expr, n], where n is the number of output decimal places. Note all *Mathematica* functions start with a capital letter and use square brackets. T

```
N[(5 + 8)/π, 10]
N[(5 + 8)/π]
```

```
4.138028520
```

```
4.13803
```

Notice in the second input line, I omitted the n. This gives output with a default number of decimals. *Mathematica* is a symbolic program. Therefore, we can make a symbol represent an expression. For example,

```
x = (2 + 5 - 15^2) (-25^3 + 8)
N[x]
```

```
3404506
```

3.40451×10^6

Notice that both lines of input were evaluated. To suppress the evaluation of a line of input, add a semicolon at the end.

```
x = (2 + 5 - 15^2) (-25^3 + 8);
N[x]
```

3.40451×10^6

Mathematica has tons of built-in mathematical functions that it will evaluate. Exp, Log, Bessel, AiryAi, Sin, Cos, Tan, etc.

▪ Algebra and Calculus

Mathematica can work with algebraic formulas. We first have to clear *x* since we previously used it to calculate something. *Mathematica* remembers these values until they are cleared. Clear[symbol_1, symbol_2, ...] clears values and definitions for the symbols.

```
Clear[x]
Expand[(2 x + 5) (1 - x^2) (2 + 5 x - x^3)]
```

$10 + 29\,x - 34\,x^3 - 12\,x^4 + 5\,x^5 + 2\,x^6$

The % symbol refers to the previous output expression. Using %% refers to the second to last output expression. This saves time.

```
Factor[%]
```

$(-1 + x)\,(1 + x)\,(2 + x)\,(5 + 2\,x)\,(-1 - 2\,x + x^2)$

Let's look at calculus. First, an integral using the "File/Palettes/Basic Calculations" palette or by typing in the necessary commands. The exponential function is Exp[] or ESC, "ee", ESC

$\int x^2\, \mathrm{Sin}[x]\, e^{-x}\, dx$

$-\frac{1}{2} e^{-x} (1 + x) ((1 + x) \mathrm{Cos}[x] + (-1 + x) \mathrm{Sin}[x])$

$$\int_0^{1.0} x^2\, \text{Sin}[x]\, e^{-x}\, dx$$

0.102468

$$\int_0^{\infty} \text{Sin}[x]\, e^{-x}\, dx$$

$$\frac{1}{2}$$

Differentiate the result of the first indefinite integral above. `D[f, x]` gives the partial derivative of the function `f` with respect to `x`. You can copy and paste the function to save time.

$$D\left[-\frac{1}{2} e^{-x} (1+x) ((1+x) \text{Cos}[x] + (-1+x) \text{Sin}[x]),\ x\right]$$

$$-\frac{1}{2} e^{-x} ((1+x) \text{Cos}[x] + (-1+x) \text{Sin}[x]) + \frac{1}{2} e^{-x} (1+x) ((1+x) \text{Cos}[x] + (-1+x) \text{Sin}[x]) - \frac{1}{2} e^{-x} (1+x) (\text{Cos}[x] + (-1+x) \text{Cos}[x] + \text{Sin}[x] - (1+x) \text{Sin}[x])$$

The expression is fairly complicated, but it should be the integrand of the integral above. We can simplify itusing the function `Simplify[expr]` or `FullSimplify[expr]`. Both try a wide rande of transformations on `expr` involving elementary and special functions and returns the simplest form it finds.

```
FullSimplify[%]
```

$$e^{-x} x^2\, \text{Sin}[x]$$

Aha! It looks like the original integrand.

We can do expansions on functions. In Modern Physics it is often important to find approximations when an analytical formula is difficult. *Mathematica* is our friend for this.

$$f = \frac{\sqrt{1 - x^2}}{1 + x};$$

```
Series[f, {x, 0, 5}]
```

$$1 - x + \frac{x^2}{2} - \frac{x^3}{2} + \frac{3 x^4}{8} - \frac{3 x^5}{8} + O[x]^6$$

$$x = \frac{v}{c};$$

```
p = %
```

$$1 - \frac{v}{c} + \frac{1}{2}\left(\frac{v}{c}\right)^2 - \frac{1}{2}\left(\frac{v}{c}\right)^3 + \frac{3}{8}\left(\frac{v}{c}\right)^4 - \frac{3}{8}\left(\frac{v}{c}\right)^5 + O\left[\frac{v}{c}\right]^6$$

```
Clear[f, x, p]
```

■ Solving Equations

`Solve[eqns, vars]` attempts to solve an equation or set of equations for the variables `vars`. Solve deals primarily with linear and polynomial equations. Solve the third order polynomial to find its roots. A double equal sign indicates and equations.

```
Solve[x^3 - 7 x^2 + 3 a x == 0, x]
```

$\left\{\{x\to 0\},\ \left\{x\to \frac{1}{2}\left(7-\sqrt{49-12\,a}\right)\right\},\ \left\{x\to \frac{1}{2}\left(7+\sqrt{49-12\,a}\right)\right\}\right\}$

Notice there is a space betwee `a` and `x`. Otherwise, *Mathematica* thinks `ax` is a single variable.

Solve a pair of simultaneous linear equations. Note that whenever you have a list of items or equations, they are placed in curly brackets and separated by commas as in the following example.

```
Solve[{a x + 2 y == 4, x - a y == 2}, {x, y}]
```

$\left\{\left\{x\to \frac{4\,(1+a)}{2+a^2},\ y\to -\frac{2\,(-2+a)}{2+a^2}\right\}\right\}$

`NSolve[lhs == rhs, var]` gives a list of numerical approximations to the roots of a polynomial equation. Note that the roots can be real or complex. For example,

```
NSolve[x^5 - 3 x^4 + x^2 - 2 x + 1 == 0, x]
```

`{{x → -1.}, {x → 0.254598 - 0.749528 i}, {x → 0.254598 + 0.749528 i}, {x → 0.541016}, {x → 2.94979}}`

`DSolve[eqn, y, x]` solves a differential equation for the function `y`, with independent variable, `x`. Solve the differential equation $y'' + 2\,\beta\,y' + \alpha\,y = A\,\mathrm{Cos}[x]$.

```
DSolve[y''[x] + 2 β y'[x] + α y[x] == A Cos[x], y[x], x]
```

$$\left\{\left\{y[x]\to e^{x\left(-\beta-\sqrt{-\alpha+\beta^2}\right)}\,C[1]+e^{x\left(-\beta+\sqrt{-\alpha+\beta^2}\right)}\,C[2]+\frac{A\,\mathrm{Cos}[x]-A\,\alpha\,\mathrm{Cos}[x]-2\,A\,\beta\,\mathrm{Sin}[x]}{\left(-1+\alpha-2\,\beta^2+2\,\beta\sqrt{-\alpha+\beta^2}\right)\left(1-\alpha+2\,\beta^2+2\,\beta\sqrt{-\alpha+\beta^2}\right)}\right\}\right\}$$

```
FullSimplify[%]
```

$$\left\{\left\{y[x]\to e^{-x\left(\beta+\sqrt{-\alpha+\beta^2}\right)}\left(C[1]+e^{2\,x\sqrt{-\alpha+\beta^2}}\,C[2]\right)+\frac{A\,((-1+\alpha)\,\mathrm{Cos}[x]+2\,\beta\,\mathrm{Sin}[x])}{(-1+\alpha)^2+4\,\beta^2}\right\}\right\}$$

■ Plotting Functions

`Plot[f, {x, xmin, xmax}]` generates a plot of `f` as a function of `x` from `xmin` to `xmax`. Plot the function $y = e^{-t}\,\mathrm{Cos}[5\,t]$.

```
Plot[e^-t Cos[5 t], {t, 0, 5}]
```



If the graph is cut off like this, you can use formatting to get it on scale.

```
Plot[ⅇ^-t Cos[5 t], {t, 0, 5}, PlotRange → Full]
```



Multiple functions can be plotted on a single graph using curly brackets again.

```
x = Sin[3 t];
y = t Sin[t];
Plot[{x, y}, {t, 0, 10}, PlotRange → Full]
```



Mathematica has 3-dimensional plotting capabilities. `Plot3D[f, {x, xmin, xmax}, {y, ymin, ymax}]` generates a 3D plot of `f` as a function of `x` and `y`. First, let's clear `x` and `y`.

```
Clear[x, y]
Plot3D[Sin[x] Cos[y], {x, 0, 3}, {y, 0, 3}]
```




```
Manipulate[Plot3D[Sin[c x] Cos[c y], {x, 0, 3}, {y, 0, 3}], {c, 0, 5}]
```



Parametric equation plots are also possible. `ParametricPlot[{fx, fy}, {t, tmin, tmax}]` produces a plot with `x` and `y` coordinates for f_x and f_y are generated as a function of `t`.

```
ParametricPlot[{t^2, Cos[t]^2}, {t, 0, 10}, AspectRatio → 1/2]
```



Similarly, the 3D parametric plot produces a 3D parametrized space curve.

```
ParametricPlot3D[{Sin[4 t], Cos[4 t], t}, {t, 0, 5}]
```



We can take this further by adding more functions to the list and another parametrization variable.

```
ParametricPlot3D[{Sin[t], Sin[2 t] Sin[q], Sin[2 t] Cos[q]}, {t, -π/2, π/2}, {q, 0, 2 π}]
```



Solving a Problem

Problem 1.91 in the textbook.
A 10 kg object is moving to the right at 0.6c. It explodes into two pieces, one of mass m_1 moving left at 0.6c and m_2 moving right at 0.8c. (a) Find m_1 and m_2. (b) Find the change in kinetic energy in this explosion.

(a) First, define the constants that are given. Then, set up equations for the relativistic corrections, γ, γ1, γ2. Use these equations to `Solve` using conservation of momentum and mass.

```
M = 10; u = 0.6 c; u1 = -0.6 c; u2 = 0.8 c;
```

$$\gamma = \frac{1}{\sqrt{1-\left(\frac{u}{c}\right)^2}};\ \gamma 1 = \frac{1}{\sqrt{1-\left(\frac{u1}{c}\right)^2}};\ \gamma 2 = \frac{1}{\sqrt{1-\left(\frac{u2}{c}\right)^2}};$$

```
Solve[{γ M u == γ1 m1 u1 + γ2 m2 u2, γ M == γ1 m1 + γ2 m2}, {m1, m2}]

{{m1 → 1.42857, m2 → 6.42857}}
```

(b) Now, we can use these masses to determine the change in kinetic energy. You'll notice there is a ` at the end of the numbers for `m1` and `m2`. This happens when you cut and paste the values. It indicates that the number has been truncated.

```
m1 = 1.428571428571429`; m2 = 6.428571428571427`;
Kfinal = (γ1 - 1) m1 c^2 + (γ2 - 1) m2 c^2;
Kinitial = (γ - 1) M c^2;
ΔK = Kfinal - Kinitial

2.14286 c^2
```

```
c = 2.998 * 10^8;
2.142857142857144` c^2
```

1.926×10^{17}

Text in Mathematica

This entire handout was written using *Mathematica*. It is very important to explain what you are doing in words when doing physics and mathematics. *Mathematica* provides a way to do this. To write text, you must place it in a text "cell". Notice that th blue brackets on the right hand side of the *Mathematica* window show different "cells". There are variety of cell types. There are input, output, text, titles, subtititles, sections, subsections, etc. To insert a text cell, place the cursor at the location that you want the cell. It should not be inside an input or output cell but between. When you have placed the cursor between cells, it will show a black line across the document. Then, you choose the type of cell that you want to use. The format that I am using at this time is "text". The easiest way to change formats is to have the toolbar displayed. The toolbar can by shown by selecting the FORMAT/-SHOW TOOLBAR menu. The pulldown window at the left side of the toolbar will display the type of cell you will use. You can change to any of the other formats by using the pulldown window to select fonts, sizes, faces, etc.

Equations Embedded in Text

The simplest way to mix text and formulas in a *Mathematica* notebook is to put each kind of material in a separate cell. Sometimes you may want to embed a formula into the text of a single cell. To do this, use the following key strokes:

CTRL-[(] or CTRL-[9] formula stuff CTRL-[)] or CTRL-[0]

For example, the average value of a function is given by $\bar{y} = \frac{1}{b-a} \int_a^b y\,(x)\, ds$. I used CTRL-[9] to start the formula, and CTRL-[0] when I had finished entering it. The formula is written in the text cell the same way it would be written or paletted in an input cell.

Further Information

Mathematica has excellent help and tutorials/demos. If you want to learn more simply go to the Help menu.

■ Exercises

1. Look up complex numbers in the *Mathematica* Help menu. Compute the real and imaginary parts of (5 + 6 i) (7 − 2 i) (4 + i). Also compute the absolute value of (5 + 6 i) (7 − 2 i) (4 + i) / (2 − 3 i).

2. Factor the following polynomial: $-2\,x^7 - 3\,x^6 + x^4 - x^3 + 3\,x + 2$.

3. Find all of the roots of the following equation: $-2\,x^7 - 3\,x^6 + x^4 - x^3 + 3\,x + 2 = 0$. There should be seven of them, including real and imaginary roots.

4. Plot the polynomial $-2\,x^7 - 3\,x^6 + x^4 - x^3 + 3\,x + 2$ for $-2 \le x \le 2$.

5. Solve and simplify the differential equation y'' + y = 1 − x Sin[x].

6. Integrate the function $\frac{\mathrm{Log}\left(x^2+1\right)}{x^2+1}$ both indefinite and for limits 0 to 10. Keep in mind that in *Mathematica* Log[] is the natural logarithm.

7. Differentiate the indefinite integral from exercise 6 to obtain the original integrand.

8. Look up the function Sum[] in *Mathematica* Help. Compute the sum of $\frac{1}{n}$ from $n = 1$ to $n = 20$. Represent the result first as an exact result, then as a decimal.

9. To what value does the sum $\sum_{n=1}^{\infty} \frac{(-1)^n}{n}$ converge?

10. Plot the parametric equations $x = t,\ y = \mathrm{Cos}\left(t^2\right)$, where $-10 \le t \le 10$.

■ Answers